

IDENTIFICATION

Product Code: MAINDEC-08-D2CA-D
Product Name: PDP-8 Teletype Punch Test
Date Created: May 5, 1965
Maintainer: Diagnostic Group
Author: R. Green
Previous Code: MAINDEC 812

1
()

()

()

1. ABSTRACT

MAINDEC-08-D2CA punches a test tape in a predetermined pattern. The tape passes directly from the Teletype punch to the Teletype reader, which checks the pattern for accuracy.

2. REQUIREMENTS

Storage

MAINDEC-08-D2CA occupies storage location 0020-0111. In addition, it uses location 0001 for an interrupt return. The RIM Loader must be in memory in locations 7756-7776.

Equipment

Standard PDP-8 with Teletype Model 33 Keyboard/Printer, including Teletype punch and reader.

3. USAGE

3.1 Loading

The RIM Loader must be in memory.

Set the Teletype to ON LINE and turn reader off. Place MAINDEC-08-D2CA RIM tape in the Teletype reader.

Set SWITCH REGISTER to 7756; press LOAD ADDRESS, then START.

Turn on reader.

When program has been read in, turn off reader, and remove program tape.

3.2 Switch Settings

SW ₀ Down	Punch leader of 377's.
Up	Punch pattern 0-377.

3.3 Start-up and/or Entry

After program has been read in, set SWITCH REGISTER to 0020.

Press LOAD ADDRESS.

Make sure that SW₀ is down (off). Press START.

When sufficient leader has been punched, place the free end of the tape in the reader (do not tear it off!) so that the first character read will be a 377.

Turn on the reader.

Set SW₀ up (on). This will cause the program to punch the test sequence.

The program will run continuously until an error occurs or until the operator stops the computer. There is no final halt.

3.4 Errors in Usage

C(MA)	Error	Cause of Error
0050	E1	Input from reader did not cause an interrupt. Possible difficulties; incorrect execution of ION; no signal from reader; trouble in the interrupt system.
0052	E2	The interrupt occurred before the instructions following the ION was executed. Probable trouble in the interrupt system.
0074	E3	The character read from tape is not what it should have been. The AC displays the character read.
0077	E3A	After the previous halt, this halt allows the operator to see what the proper character is. The correct code is displayed in the AC light.

There are no other stops in the program.

3.5 Recovery from Such Errors

Error	Recovery Procedure
E1, E2	If either of these errors occurs, perform Instruction Test 2A of MAINDEC to determine the cause of the trouble.
E3	Record the C(AC). Press CONTINUE to reach the next halt.
E3A	Record the C(AC). Compare with the C(AC) from the previous halt to determine whether information was dropped or spuriously introduced in punching. Check the tape itself for the erroneous character. To restart the test, turn off the reader; remove the tape. <ol style="list-style-type: none"> 1. Set SW₀ down (off). 2. Press CONTINUE. The test will begin again, punching leader. Follow the start-up procedure from this point. (See paragraph 3.3)

4. DESCRIPTION

4.1 Discussion

Under the control of SW_0 , MAINDEC-08-D2CA tests the performance of the Teletype punch. With SW_0 down, the program punches the code 377 to provide a leader section consisting of all holes. As this is being punched, the operator takes the free end of the tape, places it in the Teletype reader, and turns the reader on. He then sets SW_0 up, causing the program to punch the code sequence 000-377, repeating this as long as desired.

As soon as the first nonleader character is read, the program begins checking the tape for the accuracy of the punching. After each line is read, the interrupt is enabled to test its operation. If this is satisfactory, the character is checked. If an error occurs, the incorrect code is displayed in the AC and the program halts. An immediately succeeding halt displays the correct code in the AC.

If an incorrect character is encountered, the operator must restart the program. A new leader will be punched.

4.2 Examples and/or Applications

In addition to checking the accuracy of the punch, MAINDEC-08-D2CA can be used to observe other operations of the punch and reader, as explained below.

The tape drive mechanism in the punch can be checked by allowing the program to punch a few feet of leader. Examine the resulting tape for correct spacing between lines. If the tape is not advancing properly, the spacing will be irregular, or adjacent characters may overlap, resulting in extra large or odd shaped holes.

The reader's ability to scan a tape containing all holes can be observed by placing the length of leader in the reader. If it drops a bit in any line, it will automatically begin checking the tape for a pattern; this will show up immediately as an error, and the program will stop at error E3. Dropped bits will show as 0's in the AC.

5. METHODS

When the program starts, the link is set to 1. As long as the reader sees nothing but 377 codes, the link remains set. When the first pattern character appears, the link is cleared, and from then on the program will check every character for accuracy.

6. EXECUTION TIME

The program runs continuously until an error is encountered or until the operator stops the computer.

7. PROGRAM LISTING

/MAINDEC-08-D2CA: PDP-8 TELETYPE PUNCH TEST

*20

0020	6032	INIT,	KCC	/INITIALIZE COUNTERS AND STUFF
0021	7120		STL	/LEADER SIGNAL
0022	3106		DCA PCT	
0023	3107		DCA RCT	
0024	1110		TAD CTAB	/INITIALIZE INTERRUPT TEST
0025	3001		DCA I	
0026	7240		CLA CMA	
0027	6046	PUNCH,	TLS	
0030	6041	FLAGS,	TSF	/PUNCH FLAG
0031	5033		JMP .+2	
0032	5036		JMP PUNI	
0033	6031		KSF	/READER FLAG
0034	5030		JMP FLAGS	
0035	5045		JMP READ	
0036	7604	PUNI,	LAS	/TEST SW(0)
0037	7700		SMA CLA	
0040	5026		JMP PUNCH-1	/IF SW(0)=0, PUNCH LEADER
0041	1106	PPAT,	TAD PCT	/OTHERWISE, PUNCH PATTERN
0042	2106		ISZ PCT	
0043	5027		JMP PUNCH	
0044	5027		JMP PUNCH	
0045	7200	READ,	CLA	
0046	6001		ION	/TEST INTERRUPT
0047	7240		CLA CMA	/PUTS 7777 IN AC
0050	7402	E1,	HLT	/ERROR: INPUT DID NOT CAUSE AN I
0051	7450		SNA	/WAS A TAD EXECUTED?
0052	7402	E2,	HLT	/ERROR: NO, IT WASN'T.
0053	6036	R2,	KRB	/YES, IT WAS. READ BUFFER
0054	3105		DCA TEM	/AND STORE CHARACTER
0055	1111		TAD CIAB+1	/TEST FOR 377 CHARACTER.
0056	1105		TAD TEM	
0057	7040		CMA	
0060	7640		SZA CLA	/IS IT 377?
0061	7100		CLL	/NO. CLEAR LEADER SIGNAL
0062	7430		SZL	/YES. IS IT LEADER?
0063	5030		JMP FLAGS	/YES. GO WAIT FOR NEXT.

MAINDEC-08-D2CA-D

0064	1107	R3,	TAD RCT	/COMPARISON TEST.
0065	0112		AND CTAB+2	/KEEP 8 BITS OF C(RCT)
0066	2107		ISZ RCT	/INCREMENT READER COUNT
0067	7000		NOP	
0070	7161		CIA STL	/COMPLEMENT READER COUNT
0071	1105		TAD TEM	
0072	7650		SNA CLA	/IS C(TEM)-C(RCT)=0?
0073	5030		JMP FLAGS	/YES. CHARACTER IS OK.
0074	1105		TAD TEM	/NO, CHARACTER IS WRONG.
0075	7402	E3,	HLT	/ERROR: AC CONTAINS CHAR. READ
0076	7360		CLA CMA STL	
0077	1107		TAD RCT	
0100	7402	E3A,	HLT	/AC CONTAINS CORRECT CHARACTER
0101	7604		LAS	/FIND OUT WHAT
0102	7510		SPA	/TO DO?
0103	5030		JMP FLAGS	/CONTINUE
0104	5020		JMP INIT	/START OVER
0105	0000	TEM,	0	
0106	0000	PCT,	0	/PUNCH CHARACTER COUNT
0107	0000	RCT,	0	/READ CHARACTER COUNT
0110	5051	CTAB,	JMP EI+1	/INTERRUPT RETURN
0111	7400		7400	/COMPLEMENT OF 377 FOR TEST.
0112	0377		377	/8 - BIT MASK

CTAB	0110
E1	0050
E2	0052
E3	0075
E3A	0100
FLAGS	0030
INIT	0020
PCT	0106
PPAT	0041
PUNCH	0027
PUN1	0036
RCT	0107
READ	0045
R2	0053
R3	0064
TEM	0105

8. DIAGRAM

8.1 Flow Chart

